
14 Surrogate Models

The previous chapter discussed methods for producing a sampling plan. This
chapter shows how to use these samples to construct models of the objective
function that can be used in place of the real objective function. Such surrogate
models are designed to be smooth and inexpensive to evaluate so that they can be
efficiently optimized. The surrogate model can then be used to help direct the
search for the optimum of the real objective function.

14.1 Fitting Surrogate Models

A surrogate model f̂ parameterized by θ is designed to mimic the true objective
function f . The parameters θ can be adjusted to fit the model based on samples
collected from f . An example surrogate model is shown in figure 14.1.

Figure 14.1. Surrogate models ap-
proximate the true objective func-
tion. The model is fitted to the eval-
uated design points but deviates
farther away from them.

Suppose we have m design points

X =
{

x(1), x(2), . . . , x(m)
}

(14.1)

and associated function evaluations

y =
{

y(1), y(2), . . . , y(m)
}

(14.2)

For a particular set of parameters, the model will predict

ŷ =
{

f̂θ(x
(1)), f̂θ(x

(2)), . . . , f̂θ(x
(m))

}

(14.3)

Fitting a model to a set of points requires tuning the parameters to minimize
the difference between the true evaluations and those predicted by the model,
typically according to an Lp norm:1

1 It is common to use the L2 norm.
Minimizing this equation with an
L2 norm is equivalent to minimiz-
ing themean squared error at those
data points.

minimize
θ

‖y− ŷ‖p (14.4)

Equation (14.4) penalizes the deviation of the model only at the data points.
There is no guarantee that the model will continue to fit well away from observed
data, and model accuracy typically decreases the farther we go from the sampled
points.

This form of model fitting is called regression. A large body of work exists for
solving regression problems, and it is extensively studied in machine learning.2 2 K. P. Murphy, Machine Learning: A

Probabilistic Perspective. MIT Press,
2012.The rest of this chapter covers several popular surrogate models and algorithms

for fitting surrogate models to data, and concludes with methods for choosing
between types of models.

14.2 Linear Models

A simple surrogate model is the linear model, which has the form3 3 This equation may seem familiar.
It is the equation for a hyperplane.

f̂ = w0 + w⊤x θ = {w0, w} (14.5)

For an n-dimensional design space, the linear model has n + 1 parameters, and
thus requires at least n + 1 samples to fit unambiguously.

Instead of having both w and w0 as parameters, it is common to construct a
single vector of parameters θ = [w0, w] and prepend 1 to the vector x to get

f̂ = θ⊤x (14.6)

Finding an optimal θ requires solving a linear regression problem:

minimize
θ

‖y− ŷ‖2
2 (14.7)

which is equivalent to solving

minimize
θ

‖y− Xθ‖2
2 (14.8)

where X is a design matrix formed from m data points

X =









(x(1))⊤

(x(2))⊤
...

(x(m))⊤









(14.9)

© 2019 Massachusetts Institute of Technology, shared under a Creative Commons CC-BY-NC-ND license.
2022-05-22 00:25:57-07:00, revision 47fd495, comments to bugs@algorithmsbook.com



14.3. basis functions 255

function design_matrix(X)
n, m = length(X[1]), length(X)
return [j==0 ? 1.0 : X[i][j] for i in 1:m, j in 0:n]

end
function linear_regression(X, y)

θ = pinv(design_matrix(X))*y
return x -> θ⋅[1; x]

end

Algorithm 14.1. A method for con-
structing a designmatrix from a list
of design points X and a method
for fitting a surrogate model using
linear regression to a list of design
points X and a vector of objective
function values y.

Algorithm 14.1 implements methods for computing a design matrix and for
solving a linear regression problem. Several cases for linear regression are shown
in figure 14.2.

Linear regression has an analytic solution

θ = X+y (14.10)

where X+ is the Moore-Penrose pseudoinverse of X.
If X⊤X is invertible, the pseudoinverse can be computed as

X+ =
(

X⊤X
)−1

X⊤ (14.11)

If XX⊤ is invertible, the pseudoinverse can be computed as

X+ = X⊤
(

XX⊤
)−1

(14.12)

The function pinv computes the pseudoinverse of a given matrix.4 4 The function pinv uses the sin-
gular value decomposition, X =
UΣV∗, to compute the pseudoin-
verse:

X+ = VΣ+U∗

where the pseudoinverse of the di-
agonal matrix Σ is obtained by tak-
ing the reciprocal of each nonzero
element of the diagonal and then
transposing the result.

14.3 Basis Functions

The linear model is a linear combination of the components of x:

f̂ (x) = θ1x1 + · · ·+ θnxn =
n

∑
i=1

θixi = θ⊤x (14.13)

which is a specific example of a more general linear combination of basis functions

f̂ (x) = θ1b1(x) + · · ·+ θqbq(x) =
q

∑
i=1

θibi(x) = θ⊤b(x) (14.14)

Figure 14.2. Models resulting from
linear regression, whichminimizes
the square vertical distance of the
model from each point. The pseu-
doinverse produces a unique solu-
tion for any nonempty point con-
figuration.

The bottom-left subfigure
shows the model obtained for
two repeated points, in this case,
m = n + 1. Because the two entries
are repeated, the matrix X is
singular. Although X does not
have an inverse in this case, the
pseudoinverse produces a unique
solution that passes between the
two points.

In the case of linear regression, the basis functions simply extract each component,
bi(x) = xi.

Any surrogate model represented as a linear combination of basis functions
can be fit using regression:

minimize
θ

‖y− Bθ‖2
2 (14.15)

where B is the basis matrix formed from m data points:

B =









b(x(1))⊤

b(x(2))⊤
...

b(x(m))⊤









(14.16)

The weighting parameters can be obtained using the pseudoinverse

θ = B+y (14.17)

Algorithm 14.2 implements this more general regression procedure.

using LinearAlgebra
function regression(X, y, bases)

B = [b(x) for x in X, b in bases]
θ = pinv(B)*y
return x -> sum(θ[i] * bases[i](x) for i in 1 : length(θ))

end

Algorithm 14.2. A method for fit-
ting a surrogate model to a list of
design points X and corresponding
objective function values yusing re-
gression with basis functions con-
tained in the bases array.

Linear models cannot capture nonlinear relations. There are a variety of other
families of basis functions that can represent more expressive surrogate models.
The remainder of this section discusses a few common families.

14.3.1 Polynomial Basis Functions
Polynomial basis functions consist of a product of design vector components, each
raised to a power. Linear basis functions are a special case of polynomial basis
functions.

From the Taylor series expansion5 we know that any infinitely differentiable 5 Covered in appendix C.2.
function can be closely approximated by a polynomial of sufficient degree. We
can construct these bases using algorithm 14.3.

In one dimension, a polynomial model of degree k has the form

f̂ (x) = θ0 + θ1x + θ2x2 + θ3x3 + · · ·+ θkxk =
k

∑
i=0

θix
i (14.18)

Hence, we have a set of basis functions bi(x) = xi for i ranging from 0 to k.
In two dimensions, a polynomial model of degree k has basis functions of the

form
bij(x) = xi

1x
j
2 for i, j ∈ {0, . . . , k}, i + j ≤ k (14.19)

Fitting a polynomial surrogate model is a regression problem, so a polynomial
model is linear in higher dimensional space (figure 14.3). Any linear combination
of basis functions can be viewed as linear regression in a higher dimensional
space.

x

y

x x2

y

f (x)

f̂ (x)

Figure 14.3. A polynomial model
is linear in higher dimensions. The
function exists in the plane formed
from its bases, but it does not oc-
cupy the entire plane because the
terms are not independent.

polynomial_bases_1d(i, k) = [x->x[i]^p for p in 0:k]
function polynomial_bases(n, k)

bases = [polynomial_bases_1d(i, k) for i in 1 : n]
terms = Function[]
for ks in Iterators.product([0:k for i in 1:n]...)

if sum(ks) ≤ k
push!(terms,

x->prod(b[j+1](x) for (j,b) in zip(ks,bases)))
end

end
return terms

end

Algorithm 14.3. A method for con-
structing an array of polynomial
basis functions up to a degree k
for the ith component of a design
point, and a method for construct-
ing a list of n-dimensional polyno-
mial bases for terms up to degree
k.

14.3.2 Sinusoidal Basis Functions
Any continuous function over a finite domain can be represented using an infi-
nite set of sinusoidal basis functions.6 A Fourier series can be constructed for any 6 The Fourier series is also exact for

functions defined over the entire
real line if the function is periodic.integrable univariate function f on an interval [a, b]

f (x) =
θ0

2
+

∞

∑
i=1

θ
(sin)
i sin

(
2πix

b− a

)

+ θ
(cos)
i cos

(
2πix

b− a

)

(14.20)

where

θ0 =
2

b− a

∫ b

a
f (x) dx (14.21)

θ
(sin)
i =

2

b− a

∫ b

a
f (x) sin

(
2πix

b− a

)

dx (14.22)

θ
(cos)
i =

2

b− a

∫ b

a
f (x) cos

(
2πix

b− a

)

dx (14.23)

Just as the first few terms of a Taylor series are used in polynomial models, so
too are the first few terms of the Fourier series used in sinusoidal models. The
bases for a single component over the domain x ∈ [a, b] are:







b0(x) = 1/2

b
(sin)
i (x) = sin

(
2πix
b−a

)

b
(cos)
i (x) = cos

(
2πix
b−a

)
(14.24)

We can combine the terms for multidimensional sinusoidal models in the same
way we combine terms in polynomial models. Algorithm 14.4 can be used to
construct sinusoidal basis functions. Several cases for sinusoidal regression are
shown in figure 14.4.

14.3.3 Radial Basis Functions
A radial function ψ is one which depends only on the distance of a point from
some center point c, such that it can be written ψ(x, c) = ψ(‖x− c‖) = ψ(r).
Radial functions are convenient basis functions because placing a radial function
contributes a hill or valley to the function landscape. Some common radial basis
functions are shown in figure 14.5.

function sinusoidal_bases_1d(j, k, a, b)
T = b[j] - a[j]
bases = Function[x->1/2]
for i in 1 : k

push!(bases, x->sin(2π*i*x[j]/T))
push!(bases, x->cos(2π*i*x[j]/T))

end
return bases

end
function sinusoidal_bases(k, a, b)

n = length(a)
bases = [sinusoidal_bases_1d(i, k, a, b) for i in 1 : n]
terms = Function[]
for ks in Iterators.product([0:2k for i in 1:n]...)

powers = [div(k+1,2) for k in ks]
if sum(powers) ≤ k

push!(terms,
x->prod(b[j+1](x) for (j,b) in zip(ks,bases)))

end
end
return terms

end

Algorithm 14.4. The method
sinusoidal_bases_1d produces a
list of basis functions up to degree
k for the ith component of the de-
sign vector given lower bound a
and upper bound b. The method
sinusoidal_bases produces all
base function combinations up to
degree k for lower-bound vector a
and upper-bound vector b.

Figure 14.4. Fitting sinusoidal
models to noisy points.

Figure 14.5. Several radial basis
functions.

14.4. f itting noisy objective functions 263

Radial basis functions require specifying the center points. One approachwhen
fitting radial basis functions to a set of data points is to use the data points as the
centers. For a set of m points, one thus constructs m radial basis functions

bi(x) = ψ(‖x− x(i)‖) for i ∈ {1, . . . , m} (14.25)

The corresponding m×m basis matrix is always semidefinite. Algorithm 14.5 can
be used to construct radial basis functions with known center points. Surrogate
models with different radial basis functions are shown in figure 14.6.

radial_bases(ψ, C, p=2) = [x->ψ(norm(x - c, p)) for c in C] Algorithm 14.5. A method for ob-
taining a list of basis functions
given a radial basis function ψ, a
list of centers C, and an Lp norm
parameter p.

−1.5

x sin(5x)

ψ = exp(−2r2)

ψ = exp(−5r2)

ψ = exp(−10r2)

Figure 14.6. Several different Gaus-
sian radial basis functions used to
fit x sin(5x) based on 4 noise-free
samples.

14.4 Fitting Noisy Objective Functions

Models fit using regression will pass as close as possible to every design point.
When the objective function evaluations are noisy, complex models are likely to
excessively contort themselves to pass through every point. However, smoother
fits are often better predictors of the true underlying objective function.

The basis regression problem specified in equation (14.15) can be augmented
to prefer smoother solutions. A regularization term is added in addition to the
prediction error in order to give preference to solutions with lower weights. The
resulting basis regression problem with L2 regularization7 is: 7 Other Lp-norms, covered in ap-

pendix C.4, can be used as well.
Using the L1 norm will encourage
sparse solutions with less influen-
tial component weights set to zero,
which can be useful in identifying
important basis functions.

minimize
θ

‖y− Bθ‖2
2 + λ‖θ‖2

2 (14.26)

where λ ≥ 0 is a smoothing parameter, with λ = 0 resulting in no smoothing.
The optimal parameter vector is given by:8

8 The matrix (B⊤B + λI
) is not al-

ways invertible if λ = 0. However,
we can always produce an invert-
ible matrix with a positive λ.

θ =
(

B⊤B + λI
)−1

B⊤y (14.27)

where I is the identity matrix.
Algorithm 14.6 implements regression with L2 regularization. Surrogate mod-

els with different radial basis functions fit to noisy samples are shown in fig-
ure 14.7.

function regression(X, y, bases, λ)
B = [b(x) for x in X, b in bases]
θ = (B'B + λ*I)\B'y
return x -> sum(θ[i] * bases[i](x) for i in 1 : length(θ))

end

Algorithm 14.6. A method for re-
gression in the presence of noise,
where λ is a smoothing term. It re-
turns a surrogate model fitted to
a list of design points X and corre-
sponding objective function values
y using regression with basis func-
tions bases.

14.5 Model Selection

So far, we have discussed how to fit a particular model to data. This section
explains how to select which model to use. We generally want to minimize gen-
eralization error, which is a measure of the error of the model on the full design
space, including points that may not be included in the data used to train the
model. One way to measure generalization error is to use the expected squared
error of its predictions:

ǫgen = Ex∼X

[(

f (x)− f̂ (x)
)2
]

(14.28)

Figure 14.7. Several different Gaus-
sian radial basis functions used to
fit x sin(5x) with zero mean, stan-
dard deviation 0.1 error based on
10 noisy samples and radial basis
function ψ = exp(−5r2).

Of course, we cannot calculate this generalization error exactly because it requires
knowing the functionwe are trying to approximate. It may be tempting to estimate
the generalization error of a model from the training error. One way to measure
training error is to use the mean squared error (MSE) of the model evaluated on
the m samples used for training:

ǫtrain =
1

m

m

∑
i=1

(

f (x(i))− f̂ (x(i))
)2

(14.29)

However, performing well on the training data does not necessarily correspond
to low generalization error. Complex models may reduce the error on the training
set, but they may not provide good predictions in other points in the design space
as illustrated in example 14.1.9 9 A major theme in machine learn-

ing is balancing model complex-
ity to avoid overfitting the training
data. K. P. Murphy, Machine Learn-
ing: A Probabilistic Perspective. MIT
Press, 2012.

This section discusses several methods for estimating generalization error.
These methods train and test on subsets of the data. We introduce the TrainTest
type (algorithm 14.7), which contains a list of training indices and a list of test
indices. Themethod fit takes in a training set and produces amodel. Themethod
metric takes a model and a test set and produces a metric, such as the mean
squared error. The method train_and_validate (algorithm 14.7) is a utility
function for training and then evaluating a model. Although we train on subsets
of the data when estimating the generalization error, once we have decided which
model to use, we can train on the full dataset.

Consider fitting polynomials of varying degrees to evaluations of the objec-
tive function

f (x) = x/10 + sin(x)/4 + exp
(

−x2
)

Below we plot polynomial surrogate models of varying degrees using the
same nine evaluations evenly spaced over [−4, 4]. The training and gener-
alization error are shown as well, where generalization is calculated over
[−5, 5].

ǫgen = 0.187

y

The plot shows that the generalization error is high for both very low and
high values of k, and that training error decreases as we increase the polyno-
mial degree. The high-degree polynomials are particularly poor predictors
for designs outside [−4, 4].

Example 14.1. A comparison of
training and generalization error
as the degree of a polynomial sur-
rogate model is varied.

struct TrainTest
train
test

end
function train_and_validate(X, y, tt, fit, metric)

model = fit(X[tt.train], y[tt.train])
return metric(model, X[tt.test], y[tt.test])

end

Algorithm 14.7. A utility type and
method for training a model and
then validating it on ametric. Here,
train and test are lists of indices
into the training data, X is a list of
design points, y is the vector of cor-
responding function evaluations,
tt is a train-test partition, fit is a
model fitting function, and metric
evaluates a model on a test set to
produce an estimate of generaliza-
tion error.14.5.1 Holdout

train test

train(•) test( f̂ , •) generalization error estimate

Figure 14.8. The holdout method
(left) partitions the data into train
and test sets.

x

y

f

f̂

training samples
holdout samples

Figure 14.9. Poor train-test splits
can result in poor model perfor-
mance.

A simple approach to estimating the generalization error is the holdout method,
which partitions the available data into a test set Dh with h samples and a training
set Dt consisting of all remaining m − h samples as shown in figure 14.8. The
training set is used to fit model parameters. The held out test set is not used
during model fitting, and can thus be used to estimate the generalization error.
Different split ratios are used, typically ranging from 50% train, 50% test to 90%
train, 10% test, depending on the size and nature of the dataset. Using too few
samples for training can result in poor fits (figure 14.9), whereas using too many
will result in poor generalization estimates.

The holdout error for a model f̂ fit to the training set is

ǫholdout =
1

h ∑
(x,y)∈Dh

(

y− f̂ (x)
)2

(14.30)

Even if the partition ratio is fixed, the holdout error will depend on the partic-
ular train-test partition chosen. Choosing a partition at random (algorithm 14.8)
will only give a point estimate. In random subsampling (algorithm 14.9), we apply
the holdout method multiple times with randomly selected train-test partitions.
The estimated generalization error is the mean over all runs.10 Because the valida- 10 The standard deviation over all

runs can be used to estimate the
standard deviation of the esti-
mated generalization error.

tion sets are chosen randomly, this method does not guarantee that we validate
on all of the data points.

function holdout_partition(m, h=div(m,2))
p = randperm(m)
train = p[(h+1):m]
holdout = p[1:h]
return TrainTest(train, holdout)

end

Algorithm 14.8. A method for
randomly partitioning m data sam-
ples into training and holdout sets,
where h samples are assigned to
the holdout set.

function random_subsampling(X, y, fit, metric;
h=div(length(X),2), k_max=10)
m = length(X)
mean(train_and_validate(X, y, holdout_partition(m, h),

fit, metric) for k in 1 : k_max)
end

Algorithm 14.9. The random sub-
sampling method used to obtain
mean and standard deviation esti-
mates for model generalization er-
ror using k_max runs of the holdout
method.

14.5.2 Cross Validation

D1 D2 D3 D4 D5

train(••••) test( f̂ , •) generalization error estimate
train(••••) test( f̂ , •) generalization error estimate
train(••••) test( f̂ , •) generalization error estimate
train(••••) test( f̂ , •) generalization error estimate
train(••••) test( f̂ , •) generalization error estimate

generalization error µ and σ

Figure 14.10. Cross-validation par-
titions the data into equally sized
sets. Each set is the holdout set
once. Here we show 5-fold cross-
validation.

Using a train-test partition can be wasteful because our model tuning can take
advantage only of a segment of our data. Better results are often obtained by using
k-fold cross validation.11 Here, the original dataset D is randomly partitioned into k 11 Also known as rotation estimation.
setsD1,D2, . . . ,Dk of equal, or approximately equal, size, as shown in figure 14.10
and implemented in algorithm 14.10. We then train k models, one on each subset
of k− 1 sets, and we use the withheld set to estimate the generalization error. The
cross-validation estimate of generalization error is the mean generalization error
over all folds:12

12 As with random subsampling,
an estimate of variance can be ob-
tained from the standard deviation
over folds.

ǫcross-validation =
1

k

k

∑
i=1

ǫ
(i)
cross-validation

ǫ
(i)
cross-validation =

1

|D(i)
test|

∑
(x,y)∈D(i)

test

(

y− f̂ (i)(x)
)2 (14.31)

where ǫ
(i)
cross-validation and D(i)

test are the cross-validation estimate and the withheld
test set, respectively, for the ith fold.

function k_fold_cross_validation_sets(m, k)
perm = randperm(m)
sets = TrainTest[]
for i = 1:k

validate = perm[i:k:m];
train = perm[setdiff(1:m, i:k:m)]
push!(sets, TrainTest(train, validate))

end
return sets

end
function cross_validation_estimate(X, y, sets, fit, metric)

mean(train_and_validate(X, y, tt, fit, metric)
for tt in sets)

end

Algorithm 14.10. The method
k_fold_cross_validation_sets
constructs the sets needed for
k-fold cross validation on m
samples, with k ≤ m. The method
cross_validation_estimate
computes the mean of the gen-
eralization error estimate by
training and validating on the list
of train-validate sets contained
in sets. The other variables are
the list of design points X, the
corresponding objective function
values y, a function fit that trains
a surrogate model, and a function
metric that evaluates a model on
a data set.

Cross-validation also depends on the particular data partition. An exception is
leave-one-out cross-validation with k = m, which has a deterministic partition. It
trains on as much data as possible, but it requires training m models.13 Averag- 13 M. Stone, “Cross-Validatory

Choice and Assessment of Statisti-
cal Predictions,” Journal of the Royal
Statistical Society, vol. 36, no. 2,
pp. 111–147, 1974.

ing over all ( m
m/k) possible partitions, known as complete cross-validation, is often

too expensive. While one can average multiple cross-validation runs, it is more
common to average the models from a single cross-validation partition.

Cross-validation is demonstrated in example 14.2.

14.5.3 The Bootstrap
The bootstrap method14 uses multiple bootstrap samples, which consist of m indices 14 B. Efron, “Bootstrap Methods:

Another Look at the Jackknife,”The
Annals of Statistics, vol. 7, pp. 1–26,
1979.

into a dataset of size m independently chosen uniformly at random. The indices
are chosen with replacement, so some indices may be chosen multiple times and
some indices may not be chosen at all as shown in figure 14.11. The bootstrap
sample is used to fit a model that is then evaluated on the original training set. A
method for obtaining bootstrap samples is given in algorithm 14.11.

If b bootstrap samples are made, then the bootstrap estimate of the general-
ization error is the mean of the corresponding generalization error estimates
ǫ
(1)
test, . . . , ǫ

(b)
test:

ǫboot =
1

b

b

∑
i=1

ǫ
(i)
test (14.32)

=
1

m

m

∑
j=1

1

b

b

∑
i=1

(

y(j) − f̂ (i)(x(j))
)2

(14.33)

where f̂ (i) is the model fit to the ith bootstrap sample. The bootstrap method is
implemented in algorithm 14.12.

The bootstrap error in equation (14.32) tests models on data points to which
they were fit. The leave-one-out bootstrap estimate removes this source of bias by
only evaluating fitted models to withheld data:

ǫleave-one-out-boot =
1

m

m

∑
j=1

1

c−j

b

∑
i=1







(

y(j) − f̂ (i)(x(j))
)2

if jth index was not in the ith bootstrap sample
0 otherwise

(14.34)
where c−j is the number of bootstrap samples that do not contain index j. The
leave-one-out bootstrap method is implemented in algorithm 14.13.

The probability of a particular index not being in a bootstrap sample is:
(

1− 1

m

)m

≈ e−1 ≈ 0.368 (14.35)

so a bootstrap sample is expected to have on average 0.632m distinct indices from
the original dataset.

Suppose we want to fit a noisy objective function using radial basis functions
with the noise hyperparameter λ (section 14.4). We can use cross validation
to determine λ. We are given ten samples from our noisy objective function.
In practice, the objective function will be unknown, but this example uses

f (x) = sin(2x) cos(10x) + ǫ/10

where x ∈ [0, 1] and ǫ is random noise with zero mean and unit variance,
ǫ ∼ N (0, 1).

Random.seed!(0)
f = x->sin(2x)*cos(10x)
X = rand(10)
y = f.(X) + randn(length(X))/10

We will use three folds assigned randomly:
sets = k_fold_cross_validation_sets(length(X), 3)

Next, we implement our metric. We use the mean squared error:
metric = (f, X, y)->begin

m = length(X)
return sum((f(X[i]) - y[i])^2 for i in m)/m

end

We now loop through different values of λ and fit different radial basis
functions. We will use the Gaussian radial basis. Cross validation is used to
obtain the MSE for each value:
λs = 10 .^ range(-4, stop=2, length=101)
es = []
basis = r->exp(-5r^2)
for λ in λs

fit = (X, y)->regression(X, y, radial_bases(basis, X), λ)
push!(es,

cross_validation_estimate(X, y, sets, fit, metric)[1])
end

The resulting curve has a minimum at λ ≈ 0.2.

Example 14.2. Cross valida-
tion used to fit a hyperparameter.

train(•) test( f̂ , •) generalization error estimate

Figure 14.11. A single bootstrap
sample consists of m indices into
the dataset sampled with replace-
ment. A bootstrap sample is used
to train amodel, which is evaluated
on the full dataset to obtain an esti-
mate of the generalization error.

bootstrap_sets(m, b) = [TrainTest(rand(1:m, m), 1:m) for i in 1:b] Algorithm 14.11. A method for ob-
taining b bootstrap samples, each
for a data set of size m.

function bootstrap_estimate(X, y, sets, fit, metric)
mean(train_and_validate(X, y, tt, fit, metric) for tt in sets)

end

Algorithm 14.12. A method for
computing the bootstrap general-
ization error estimate by training
and validating on the list of train-
validate sets contained in sets. The
other variables are the list of design
points X, the corresponding objec-
tive function values y, a function
fit that trains a surrogate model,
and a function metric that evalu-
ates a model on a data set.

function leave_one_out_bootstrap_estimate(X, y, sets, fit, metric)
m, b = length(X), length(sets)
ε = 0.0
models = [fit(X[tt.train], y[tt.train]) for tt in sets]
for j in 1 : m

c = 0
δ = 0.0
for i in 1 : b

if j ∉ sets[i].train
c += 1
δ += metric(models[i], [X[j]], [y[j]])

end
end
ε += δ/c

end
return ε/m

end

Algorithm 14.13. A method for
computing the leave-one-out boot-
strap generalization error estimate
using the train-validate sets sets.
The other variables are the list of
design points X, the correspond-
ing objective function values y, a
function fit that trains a surrogate
model, and a function metric that
evaluates a model on a data set.

Unfortunately, the leave-one-out bootstrap estimate introduces a new bias
due to the varying test set sizes. The 0.632 bootstrap estimate15 (algorithm 14.14)

15 The 0.632 bootstrap estimate was
introduced in B. Efron, “Estimat-
ing the Error Rate of a Predic-
tion Rule: Improvement on Cross-
Validation,” Journal of the Ameri-
can Statistical Association, vol. 78,
no. 382, pp. 316–331, 1983. A vari-
ant, the 0.632+ bootstrap estimate,
was introduced in B. Efron and
R. Tibshirani, “Improvements on
Cross-Validation: The .632+ Boot-
strap Method,” Journal of the Amer-
ican Statistical Association, vol. 92,
no. 438, pp. 548–560, 1997.

alleviates this bias:

ǫ0.632-boot = 0.632ǫleave-one-out-boot + 0.368ǫboot (14.36)

function bootstrap_632_estimate(X, y, sets, fit, metric)
models = [fit(X[tt.train], y[tt.train]) for tt in sets]
ϵ_loob = leave_one_out_bootstrap_estimate(X,y,sets,fit,metric)
ϵ_boot = bootstrap_estimate(X,y,sets,fit,metric)
return 0.632ϵ_loob + 0.368ϵ_boot

end Algorithm 14.14. A method for
obtaining the 0.632 bootstrap es-
timate for data points X, objective
function values y, number of boot-
strap samples b, fitting function
fit, and metric function metric.

Several generalization estimation methods are compared in example 14.3.

Consider ten evenly spread samples of f (x) = x2 + ǫ/2 over x ∈ [−3, 3],
where ǫ is zero-mean, unit-variance Gaussian noise. We would like to test
several different generalization error estimationmethodswhen fitting a linear
model to this data. Our metric is the root mean squared error, which is the
square root of the mean squared error.

The methods used are the holdout method with eight training samples,
five-fold cross validation, and the bootstrap methods each with ten bootstrap
samples. Each method was fitted 100 times and the resulting statistics are
shown below.

0 2 4 6 8 10 12

0.632 Bootstrap
Leave-One-Out Bootstrap

Bootstrap
5-Fold Cross Validation

Holdout

root mean squared error

Example 14.3. A comparison
of generalization error estimation
methods. The vertical lines in the
box and whisker plots indicate the
minimum, maximum, first and
third quartiles, and median of ev-
ery generalization error estimation
method among 50 trials.

14.6 Summary

• Surrogate models are function approximations that can be optimized instead
of the true, potentially expensive objective function.

• Many surrogate models can be represented using a linear combination of basis
functions.

• Model selection involves a bias-variance tradeoff between models with low
complexity that cannot capture important trends and models with high com-
plexity that overfit to noise.

• Generalization error can be estimated using techniques such as holdout, k-fold
cross validation, and the bootstrap.

14.7 Exercises

Exercise 14.1. Derive an expression satisfied by the optimum of the regression
problem equation (14.8) by setting the gradient to zero. Do not invert anymatrices.
The resulting relation is called the normal equation.

Exercise 14.2. When would we use a more descriptive model, for example, with
polynomial features, versus a simpler model like linear regression?

Exercise 14.3. A linear regression problem of the form in equation (14.8) is not
always solved analytically, and optimization techniques are used instead. Why is
this the case?

Exercise 14.4. Suppose we evaluate our objective function at four points: 1, 2,
3, and 4, and we get back 0, 5, 4, and 6. We want to fit a polynomial model
f (x) = ∑

k
i=0 θix

i. Compute the leave-one-out cross validation estimate of the
mean squared error as k varies between 0 and 4. According to this metric, what is
the best value for k, and what are the best values for the elements of θ?

